

## Exercise Lecture XII Variational Monte Carlo (VMC)

### 1. Quantum oscillator in 1D: variational approach (see Fig. 1)

Consider the hamiltonian of the quantum harmonic oscillator  $\mathcal{H} = p^2/2 + x^2/2$  (having chosen  $m = k = \hbar = 1$ ). We want to solve it numerically with VMC. To this purpose consider two different choices for the trial wavefunction:

$$\psi(x) = \begin{cases} B(a^2 - x^2), & \text{for } |x| < a; \\ 0, & \text{for } |x| > a. \end{cases} \quad (1.a)$$

or:

$$\psi(x) = Ae^{-\beta x^2} \quad (1.b)$$

with  $a$  and  $\beta$  variational parameters,  $A$  and  $B$  proper normalization constants.

(a) Calculate

$$\langle E \rangle = \langle E_L \rangle = \frac{1}{M} \sum_{i=1}^M E_L(x_i),$$

with

$$E_L(x) = \frac{\mathcal{H}\psi(x)}{\psi(x)} \quad (\text{local energy})$$

and with the  $x_i$  distributed according  $\psi(x)^2$ . Using the trial wavefunction (1.b), remind the exercise concerning the Metropolis sampling of a gaussian function, Lecture VII; see the code `metropolis_gaussian.f90`. See `metropolis_parabola.f90` for (1.a).

(b) Verify numerically and analytically that the variational solution gives in the two cases:

(1.a):  $a = (35/2)^{1/4} \approx 2.0453$ ;  $\langle E \rangle = 0.6$

(1.b):  $\beta = \frac{1}{2}$ ;  $\langle E \rangle = 0.5$

*(In this case the exact analytic solution is known,  $E_0 = 0.5$ ; however, in case (1.a) the minimum of  $\langle E \rangle$  as a function of  $a$  is rather flat: use steps of 0.01 for  $a$  and  $n = 100000$  Metropolis accumulation steps to appreciate the minimum)*

(c) Instead of solving the problem by minimizing  $\langle E_L \rangle$ , solve it minimizing the variance:

$$\sigma^2 = \langle E_L^2 \rangle - \langle E_L \rangle^2;$$

you should observe that in the case (1.a) the minimum of the variance is positive, whereas in the case (1.b) is zero.

*(Important: note the usefulness of the “zero variance property”; the EXACT minimum of  $\sigma^2$  is 0 is the class of the trial wavefunctions contains the exact result, whereas in general the minimum of  $\langle E \rangle$  is not known!)*

### 2. Anharmonic quantum oscillator 1D (see Fig. 2)

(a) Consider also an anharmonic part in the potential, i.e.  $V(x) = x^2/2 + bx^4$ . Set for instance  $b = 1/8$ , make a plot of  $V(x)$ . Use the first order perturbation theory to calculate the changes at the lowest order in energy of the ground state due to the anharmonic term. Choose a reasonable form for the trial wavefunction, e.g. (1.b), and calculate  $\langle E \rangle$  with VMC. Compare the result obtained with the one obtained using the perturbation theory.

### 3. Hydrogen atom. (Optional)

- (a) We want to find numerically the ground state of the H atom. We want to find numerically the ground state. The hamiltonian is:  $\mathcal{H} = p^2/(2m) - e^2/r$ ; It is convenient to use atomic units ( $\hbar = 1$ ,  $m_e = 1/2$ ,  $e^2=2$ , and therefore to measure all length in terms of Bohr radius,  $a_0 = \hbar^2/(me^2)$ , the energies in Rydberg,  $Ry = me^4/(2\hbar^2)$ ). Consider the trial wavefunction  $\psi(\mathbf{r}) = \psi(r) = e^{-r/a}$ , where  $a$  is a variational parameter. calculate the optimal value of  $ak$  and the corresponding energy.

NOTE no.1: you may reduce the problem (spherical part) in a 1D problem. Pay attention that

$$\langle E_L \rangle = \frac{\int \psi^2(r) E_L(r) dr}{\int \psi^2(r) dr} = \frac{\int \psi^2(r) E_L(r) 4\pi r^2 dr}{\int \psi^2(r) 4\pi r^2 dr}$$

and therefore the probability for the sampling is  $r^2\psi^2(r)$  and not  $\psi^2(r)$ .

NOTE no.2: Pay attention close to the origin!

- (b) Do the same in 3D, considering the isotropy of the trial move (The displacement is a vector!)

```
!cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
! metropolis_gaussian.f90
!
! METROPOLIS sampling of several physical observables for the
! hamiltonian:      h = -1/2 \nabla^2 + 1/2 x^2),
! comparison exact expected results with numerical results
! on psi^2(x), with psi(x) = exp(-x^2/(4\sigma^2))
! \sigma=1 => psi^2(x) = constant * standard gaussian
! P(x) = exp(-x**2/(2*\sigma**2))/sqrt(2*pi*\sigma**2)
!cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
```

```
program metropolis_gaussian
  implicit none
  integer, parameter :: dp=selected_real_kind(13)
  integer :: i,n
  real(kind=dp):: sigma,etot,ekin,epot,rnd,ekinL,epotL,etot2
  real (kind=dp) :: pigr,pi2b,var,beta,var_th
  real(kind=dp):: x,x1,x2,xp,delta,expx,expxp,p,acc
  character(len=13), save :: format1 = "(a7,2x,2f9.5)"
  open(unit=7,file='e_var_gauss.dat',position='append')

  pigr = 2*asin(1.0_dp)
  acc = 0.0_dp
  x1 = 0.0_dp
  x2 = 0.0_dp
  ekin = 0.0_dp
  epot = 0.0_dp
```

```

etot2= 0.0_dp
print*, "n, sigma (remember: beta = 1 / (4*sigma**2)), x0, delta"
read*, n,sigma,x,delta
beta = 1 / (4*sigma**2)
! call random_seed(put=seed)

do i=1,n
  ekinL = - 0.5_dp * ((x/(2*sigma**2))**2 - 1/(2*sigma**2))
  epotL = 0.5_dp * x**2
  ekin = ekin + ekinL
  epot = epot + epotL
  etot = ekin + epot
  etot2 = etot2 + (ekinL + epotL)**2
  x1 = x1 + x
  x2 = x2 + x**2
  !cccccccccccccccccccccccccccccccccccccccc
  expx = - x**2 / (2*sigma**2) !
  call random_number(rnd) !
  xp = x + delta * (rnd-0.5_dp) !
  expxp = - xp**2 / (2*sigma**2) ! metropolis
  p = exp (expxp-expx) ! algorithm
  call random_number(rnd) !
  if (p > rnd) then !
    x = xp !
  !cccccccccccccccccccccccccccccccccccccccc
  acc=acc+1.0_dp
endif
enddo

var_th = 1._dp/(32*beta**2)+beta**2/2-1._dp/4

write(unit=*,fmt=*)"acceptance ratio = ",acc/n
write(unit=*,fmt=*)"# Results (simulation vs. exact results):"
write(unit=*,fmt=format1)"etot = ",etot/n,1.0_dp/(8.0_dp*sigma**2)&
+0.5_dp*sigma**2
write(unit=*,fmt=format1)"ekin = ",ekin/n,1.0_dp/(8.0_dp*sigma**2)
write(unit=*,fmt=format1)"epot = ",epot/n,0.5_dp*sigma**2
write(unit=*,fmt=format1)"evar = ",etot2/n-(etot/n)**2,var_th
write(unit=*,fmt=format1)"<x> = ",x1/n,0.0_dp
write(unit=*,fmt=format1)"<x^2>= ",x2/n,sigma**2

write(7,*)sigma,etot/n,sqrt(abs(etot2/n-(etot/n)**2)),etot2/n-(etot/n)**2

close(7)

end program metropolis_gaussian

```




Figure 1: **Harmonic oscillator in 1D with VMC (es. 2)**: Ground state Eigenvalue and Eigenstate with two different trial wavefunctions.




Figure 2: **Harmonic oscillator in 1D with VMC (es. 3)**: Ground state Eigenstate and eigenvalue with two different trial wavefunctions, and comparison with the analytic solution obtained within the first order perturbation theory.